

Introdução à Informática

SE										
=MÍNIMO(G3*SE(G3<=868,29;7,65%;SE(G3										
	A	B	C	D		E	F	G	H	
1	Nome	Salário base	Dep. IRRF	Horas-extras		Vir. das horas extras	Salário bruto	INSS	Ou desc	
2				70%	100%					
3	Mônica	6.500,00	2	7,0	10,0	942,50	7.442,50	=MÍNIMO(G3*		
4	Cebolinha	386,00	4	15,0	6,0	65,80	451,80	34,56		
5	Cascão	580,00	2	6,0	7,0	63,80	643,80	49,25		

Estudo dirigido de fórmulas para Planilhas de Cálculo

José Luís Carneiro



Introdução

Este exercício trata do desenvolvimento de uma planilha para o cálculo de acréscimos e descontos uma folha de pagamento *simplificada*. Seu objetivo é demonstrar o uso prático de algumas fórmulas básicas de planilhas de cálculos.

Cabe observar que as regras para deduções e acréscimos seguem as determinações do Governo Federal mas apresentam algumas simplificações de cunho didático. Por exemplo: a regra de dependência para cálculo do Salário-família difere da regra de dependência para cálculo do Imposto de Renda, entretanto, para simplificar, neste exercício foram considerados que todos os dependentes seguem a mesma regra.

O problema

Em uma determinada empresa sem a folha de pagamento informatizada, tornou-se necessária uma ferramenta que facilitasse o cálculo dos acréscimos e descontos no salário dos funcionários.

Os acréscimos possíveis são:

- Horas-extras simples – acréscimo de 70% sobre o valor da hora trabalhada do funcionário;
- Horas-extras especiais – acréscimo de 100% sobre o valor da hora trabalhada do funcionário;
- Salário-família – os funcionários têm direito a um adicional por dependente baseado em sua renda bruta mensal:

Salário-base (em R\$)	Adicional por dependente (em R\$)
Até 449,93	23,08
De 449,94 até 676,27	16,26

Os descontos possíveis são:

- INSS:

Renda bruta mensal (em R\$)	Alíquota (%)
Até R\$ 868,29	7,65
De R\$ 868,30 a R\$ 1.040,00	8,65
De R\$ 1.040,01 a R\$ 1.447,14	9,00
A partir de R\$ 1.447,15	11,00

Desconto *máximo* de R\$ 318,37.

- IRRF:

Base de cálculo mensal (em R\$)	Alíquota (%)	Dedução (em R\$)
Até 1.313,69	–	–
De 1.313,70 até 2.625,12	15,0	197,05
A partir de 2.625,13	27,5	525,19

Dedução por dependente: R\$ 132,05.

As colunas

A planilha seguirá o modelo abaixo:

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Nome	Salário base	Dep. IRRF	Horas extras		Valor horas extras	Salário bruto	INSS	Outros descontos	IRRF		Salário família	Salário líquido
2				70%	100%					Base	Desconto		
3	Ubirajara	6.500,00	2	7,0	10,0								

Solução

É possível resolver essa planilha usando apenas as funções básicas no uso de planilhas eletrônicas:

- SE(...)
- MÁXIMO(...)
- MÍNIMO(...)

Valor das horas-extras (célula F3)

1. Calcula-se o valor da hora trabalhada de cada funcionário, dividindo-se seu Salário-base pelo número de horas no mês (o limite máximo permitido pelo Governo Federal é 220 horas por mês):

=B3/220

2. Para obter o valor da hora-extra simples, calcula-se acréscimo de 70% sobre o valor da hora trabalhada:

=B3/220*170%

3. Multiplica-se pelo número de horas-extras simples trabalhadas:

=B3/220*170%*D3

4. Faz-se cálculo análogo para as horas-extras especiais (acréscimo de 100%):

=B3/220*170%*D3 + B3/220*200%*E3

Salário-bruto (célula G3)

Somam-se o Salário-base e as horas-extras:

=B3+F3

INSS (célula H3)

O desconto do INSS é o produto entre o Salário-base e a alíquota:

=B3 * alíquota

A alíquota é determinada por quatro faixas de valores. Como cada faixa é determinada por uma condição, é possível usar a função SE(...). Uma vez que esta função tem apenas duas saídas, para obter os *quatro* resultados possíveis, serão necessárias *três* funções SE(...) aninhadas:

1. Primeira faixa – caso o Salário-base seja menor ou igual a R\$ 868,29, a alíquota será de 7,65%:

SE (B3<=868,29; 7,65%;

2. Segunda faixa – o Salário-base que não se enquadrar na condição anterior deve ser testados. Inclui-se, então, uma nova função SE(...): caso o salário-base seja menor ou igual a R\$ 1.040,00, a alíquota será de 8,65%:

SE (B3<=868,29; 7,65%; SE (B3<=1040; 8,65%;

3. Terceira faixa – o Salário-base que não se enquadrar na condição anterior deve ser testados. Inclui-se, então, uma nova função SE(...): caso o Salário-base seja menor ou igual a R\$ 1.447,14, a alíquota será de 9%:

SE (B3<=868,29; 7,65%; SE (B3<=1040; 8,65%; SE (B3<=1447,14; 9%;

4. Quarta-faixa – o Salário-base que não se enquadrar na condição anterior terá alíquota de 11%:

SE (B3<=868,29; 7,65%; SE (B3<=1040; 8,65%; SE (B3<=1447,14; 9%; 11%)))

Até o momento, a fórmula desta célula é:

=B3*SE (B3<=868,29; 7,65%; SE (B3<=1040; 8,65%; SE (B3<=1447,14; 9%; 11%)))

Entranto, segundo as regras do Governo Federal, o desconto máximo permitido é R\$ 318,87. Assim, será necessário acrescentar a função MÍNIMO(...), que retornará o *menor* de dois valores – o limite (318,87) ou o valor calculado:

=MÍNIMO (B3*SE (B3<=868,29;7,65%;SE (B3<=1040;8,65%;SE (B3<=1447,14;9%;11%))); 318,37)

Outros descontos (célula I3)

Esta célula será *preenchida* com outros descontos que o funcionário possa sofrer (pensão alimentícia, adiantamentos, etc.).

IRRF – Base de cálculo (célula J3)

A base de cálculo do IRRF será o Salário-bruto menos o desconto do INSS, os Outros descontos e a dedução por dependente permitida pelo Governo Federal (R\$ 132,05) multiplicada pelo número de dependentes:

=G3-H3-I3-(132,05*C3)

Entretanto, para evitar que funcionários com salário muitos dependentes fiquem com base de cálculo negativa, será necessário acrescentar a função MÁXIMO(...), que retornará o *maior* de dois valores – o desconto ou zero:

=MÁXIMO (G3-H3-I3-(132,05*C3); 0)

IRRF – Desconto (célula K3)

O desconto do IRRF é o produto entre a base de cálculo e a alíquota menos o desconto-padrão dado pelo Governo Federal:

=J3 * alíquota - desconto-padrão

A alíquota é determinada por três faixas de valores. Como cada faixa é determinada por uma condição, é possível usar a função SE(...). Uma vez que esta função tem apenas duas saídas, para obter os *três* resultados possíveis, serão necessárias *duas* funções SE(...) aninhadas:

1. Primeira faixa – caso a base de cálculo seja menor ou igual a R\$ 1.313,69, a alíquota será de 0% (contribuinte isento):

SE (J3<=1313,69;0%;

2. Segunda faixa – a base de cálculo que não se enquadrar na condição anterior deve ser testada. Inclui-se, então, uma nova função SE(...): caso a base de cálculo seja menor ou igual a R\$ 2.625,12, a alíquota será de 15%:

SE (J3<=1313,69;0%;SE (J3<=2625,12;15%;

3. Terceira faixa – a base de cálculo que não se enquadrar na condição anterior terá alíquota de 27,5%:

SE (J3<=1313,69;0%;SE (J3<=2625,12;15%;27,5%))

Até o momento, a fórmula desta célula é:

=J3*SE (J3<=1313,69;0%;SE (J3<=2625,12;15%;27,5%))

É necessário fazer testes semelhantes para determinar o desconto-padrão a ser aplicado:

1. Primeira faixa – caso a base de cálculo seja menor ou igual a R\$ 1.313,69, o desconto-padrão será zero (contribuinte isento):

SE (J3<=1313,69;0;

2. Segunda faixa – a base de cálculo que não se enquadrar na condição anterior deve ser testada. Inclui-se, então, uma nova função SE(...): caso a base de cálculo seja menor ou igual a R\$ 2.625,12, o desconto-padrão será de R\$ 197,05:

SE (J3<=1313, 69;0;SE (J3<=2625, 12;197, 05;

4. Terceira faixa – a base de cálculo que não se enquadrar na condição anterior terá desconto-padrão de R\$ 525,19:

SE (J3<=1313, 69;0;SE (J3<=2625, 12;197, 05;525, 19))

Assim, a fórmula desta célula passa a ser:

=J3*SE (J3<=1313, 69;0%;SE (J3<=2625, 12;15%;27, 5%)) -
SE (J3<=1313, 69;0;SE (J3<=2625, 12;197, 05;525, 19))

Salário-família (célula L3)

O Salário-família é o produto entre o número de dependentes e o adicional por dependente determinado pelo Governo Federal:

=C3 * adicional por dependente

O adicional por dependente é calculado com base no Salário-base do funcionário, classificado em duas faixas. Como cada faixa é determinada por uma condição, é possível usar a função SE(...). Uma vez que esta função tem apenas duas saídas, para obter os *três* resultados possíveis, serão necessárias *duas* funções SE(...) aninhadas:

1. Primeira faixa – caso o Salário-base seja menor ou igual a R\$ 449,93, o adicional por dependente será de R\$ 23,08:

=SE (B3<=449, 93;23, 08;

2. Segunda faixa – o Salário-base que não se enquadrar na condição anterior deve ser testado. Inclui-se, então, uma nova função SE(...): caso o Salário-base seja menor ou igual a R\$ 676,27, o adicional por dependente será de R\$ 16,26:

=SE (B3<=449, 93;23, 08;SE (B3<=676, 27;16, 26;

3. Terceira faixa – o Salário-base que não se enquadrar na condição anterior terá adicional por dependente de zero (sem direito a Salário-família):

=SE (B3<=449, 93;23, 08;SE (B3<=676, 27;16, 26;0))

Assim, a fórmula para esta célula será:

=C3*SE (B3<=449, 93;23, 08;SE (B3<=676, 27;16, 26;0))

Salário líquido (célula M3)

Será igual ao Salário-bruto menos o desconto do INSS, os Outros descontos e o IRRF e acrescido do Salário-família:

=G3-H3-I3-K3+L3

Solução alternativa

É possível, mais prático e mais previdente substituir a função SE(...) pela função PROCV(...) permitindo que a planilha de cálculo procure automaticamente, em tabelas auxiliares (representações das tabelas de INSS, IRRF e Salário-família apresentadas anteriormente), faixas correspondentes a valores como Salário-bruto, Salário-base e base de cálculo, retornando as alíquotas, descontos e adicionais adequados.

A função PROCV(...)

A função PROCV(...) procura por um valor num intervalo de células (uma matriz). Uma vez encontrada a linha com a faixa de valores correspondente, uma das suas células (indicada por um índice) é retornada.

Seu funcionamento é similar a procurar por um valor numa tabela usando uma coluna como índice de pesquisa e as demais como dados desejados.

Sua sintaxe é:

=PROCV(Critério de pesquisa;Matriz;Índice;Ordem de classificação)

Onde:

Critério de pesquisa é o valor procurado na primeira coluna da matriz.

Matriz é o intervalo que contém os dados pesquisados, deve abranger pelo menos duas colunas.

Índice é o número da coluna da matriz que contém o valor a ser retornado (a primeira coluna corresponde ao índice 1).

Ordem de classificação é um parâmetro *opcional* que indica se a primeira coluna do vetor deve ser classificada em ordem crescente. Insira o valor Booleano FALSO se não desejar que a primeira coluna seja classificada em ordem crescente. As colunas classificadas podem ser pesquisadas mais rapidamente e a função as considerará como uma faixa de valores, retornando *sempre* um valor (mesmo que o valor pesquisado não possua um correspondente exato).

Folhas de cálculo auxiliares

Os programas de planilha de cálculo mais modernos permitem diversas "folhas de cálculo" num mesmo arquivo de planilha de cálculo. As folhas de cálculo podem ser observadas no parte inferior da planilha:



Portanto, por razões de organização, as tabelas auxiliares serão criadas numa folha auxiliar, devidamente renomeada (usando o botão da direita do mouse) para "Tabelas".

A folha Tabelas seguirá o modelo abaixo:

	A	B	C	D
1		Renda bruta mensal	Alíquota	
2	I N S S	0,00	7,65%	
3		868,30	8,65%	
4		1.040,01	9,00%	
5		1.447,15	11,00%	
6		Desconto máximo	318,37	
7				
8		Base de cálculo	Alíquota	Dedução-padrão
9	I R R F	0,00	0,0%	0,00
10		1.313,70	15,0%	197,05
11		2.625,13	27,5%	525,19
12		Dedução por dependente		
13				
14	F a m í l i a	Salário-base	Adicional	
15		0,00	23,08	
16		449,94	16,26	
17		676,28	0,00	

As fórmulas afetadas

De posse destas novas ferramentas (folhas de cálculo auxiliares e função PROCV(...)), verifica-se que as alterações, na folha de cálculo principal, serão nas células INSS (célula H3), IRRF – Base de cálculo (célula J3), IRRF – Desconto (célula K3) e Salário-família (célula L3).

INSS (célula H3)

O desconto do INSS é o produto entre o Salário-base e a alíquota:

$$=B3 * alíquota$$

A alíquota é determinada por quatro faixas de valores. Será usada a função PROCV(...) sobre a tabela auxiliar contendo os dados do INSS.

A tabela correspondente ao INSS está localizada na folha de cálculo Tabelas, da célula B2 até a célula C5. A nomenclatura para essas coordenadas é:

Tabelas.B2:C5

Como essas coordenadas serão as mesmas para todas as referências a essa tabela, deve-se usar Referências Absolutas. Portanto as coordenadas deverão ser escritas como:

Tabelas.\$B\$2:\$C\$5

Assim, a fórmula para essa célula, até o momento é:

$$=B3*PROCV(B3;Tabelas.B2:C5;2)$$

Entranto, segundo as regras do Governo Federal, o desconto máximo permitido é R\$ 318,87. Assim, será necessário acrescentar a função MÍNIMO(...), que retornará o *menor* de dois valores – o limite (localizado em Tabelas.C6 – observar a referência absoluta) ou o valor calculado:

$$=MÍNIMO(B3*PROCV(B3;Tabelas.B2:C5;2);Tabelas.C6)$$

IRRF – Base de cálculo (célula J3)

A base de cálculo do IRRF será o Salário-bruto menos o desconto do INSS, os Outros descontos e a dedução por dependente permitida pelo Governo Federal (localizada em Tabelas.D12, novamente com referência absoluta) multiplicada pelo número de dependentes:

$$=MÁXIMO(G3-H3-I3-(Tabelas.D12*C3);0)$$

IRRF – Desconto (célula K3)

O desconto do IRRF é o produto entre a base de cálculo e a alíquota menos o desconto-padrão dado pelo Governo Federal:

$$=J3 * alíquota - desconto-padrão$$

A alíquota é determinada por três faixas de valores. Será usada a função PROCV(...) sobre a tabela auxiliar contendo os dados do IRRF (localizada em Tabelas.B9:D11 – não esquecendo o uso de referências absolutas):

PROCV(J3;Tabelas.\$B\$9:\$D\$11;2)

O desconto-padrão é determinado de forma análoga, mudando apenas o índice da coluna retornada que passa a ser 3:

PROCV(J3;Tabelas.\$B\$9:\$D\$11;3)

Portanto, a nova fórmula para o cálculo do desconto do IRRF será:

$$=J3*PROCV(J3;Tabelas.B9:D11;2)-PROCV(J3;Tabelas.B9:D11;3)$$

Salário-família (célula L3)

O Salário-família é o produto entre o número de dependentes e o adicional por dependente determinado pelo Governo Federal:

$$=C3 * adicional por dependente$$

O adicional por dependente é calculado com base no Salário-base do funcionário, classificado em duas faixas. Será usada a função PROCV(...) sobre a tabela auxiliar contendo os dados do Salário-família (localizada em Tabelas.B15:C17 – sempre usando referências absolutas):

PROCV(B3;Tabelas.\$B\$15:\$C\$17;2)

Substituindo na fórmula geral, tem-se:

=C3 * PROCV(B3;Tabelas.\$B\$15:\$C\$17;2)

Conclusão

Como pôde-se observar, o uso da função PROCV(...) simplificou bastante as fórmulas usadas. Esse método também concentrou as alterações numa folha de cálculo auxiliar, permitindo a atualização da planilha sem alteração nas fórmulas, em caso de mudanças da legislação.

Há diversas maneiras de resolver um problema usando planilhas de cálculo. Desde o uso de funções básicas, até o uso de funções mais rebuscadas que permitam uma ferramenta mais versátil e com fórmulas mais fáceis de entender e manter.

O objetivo deste exercício não foi esgotar todos as variações possíveis das funções usadas, mas apenas apresentar aos alunos uma aplicação prática das funções apresentadas. *É importante lembrar, porém, que esta planilha tem algumas simplificações adotadas por questões de simplificação didática.*

Diversas outras melhorias podem ser implementadas neste exemplo, gerando ferramentas cada vez mais completas para a confecção de folhas de pagamento. Uma sugestão seria a contagem de horas extras ao informarem-se os horários de entrada e saída...

Referências bibliográficas

CARNEIRO, José Luís. **Introdução às fórmulas para Planilhas de Cálculo.** Disponível em <http://www.jlcarneiro.com/wp-content/uploads/File/fssa_inf/formulas.doc>. Acesso em: maio 2007.